
' -----[ Title ]----------------------------------------------------------

' PS2 Control - DisplayValuesOfButtonsAndJoysticks.bs2

'

'

' File...... PSX_Demo.BS2

' Purpose... PlayStation Controller Interface

' Author.... Jon Williams

' E-mail.... jwilliams@parallax.com

' Started...

' Updated... 17 JUL 2003

'

'

' {$STAMP BS2}

' {$PBASIC 2.5}

'

' -----[ Program Description ]---------------------------------------------

'

' The purpose of this program is to show the values of all the buttons and

' the joysticks. This information can be used later on to set up a MATLAB

' program that can take these values (and the buttons these values

' corrospond to) and converts them into usefull informaiton. After finding

' which values corrospond to which bottons pressed you can enter this

' information into the diagram located directly below the main loop.

'

' This program demonstrates the essential interface between the BASIC

' Stamp and a Sony PlayStation (or compatible) game controller. This

' code assumes that the clock signal is inverted between the Stamp and

' the controller to allow simpler [less sophisticated] interface with

' SHIFTOUT and SHIFTIN.

'

' Note: The interface and portions of code are based on previous work by

' Aaron Dahlen.

'

' -----[ Variables/Constants/Pins ]---------------------------------------

FreqDetectable    CON    3000

Piezospeaker      PIN    4                'Speaker

' For PS2 controler

' -----[ I/O Definitions ]-------------------------------------------------

PsxClk    PIN 9             ' PSX joystick interface

PsxAtt    PIN 8

PsxCmd    PIN 7

PsxDat    PIN 6

' -----[ Constants ]-------------------------------------------------------

Inverted  CON 1             ' inverted clock signal

Direct    CON 0             ' no inverter in clock line

ClockMode CON Inverted

' -----[ Variables ]-------------------------------------------------------

idx       VAR Nib           ' loop counter

psxOut    VAR Byte          ' byte to controller

psxIn     VAR Byte          ' byte from controller

' joystick packet

psxID     VAR Byte          ' controller ID

psxThumbL VAR Byte          ' left thumb buttons

psxThumbR VAR Byte          ' right thumb buttons

psxStatus VAR Byte          ' status ($5A)

psxJoyRX  VAR Byte          ' r joystick - X axis

psxJoyRY  VAR Byte          ' r joystick - Y axis

psxJoyLX  VAR Byte          ' l joystick - X axis

psxJoyLY  VAR Byte          ' l joystick - Y axis

' -----[ Initialization ]--------------------------------------------------

FREQOUT Piezospeaker, 2000, FreqDetectable          'Signal program start/reset.

Setup:

HIGH PsxAtt                 ' deselect PSX controller

OUTPUT PsxCmd

PsxClk = ~ClockMode         ' release clock

OUTPUT PsxClk               ' make clock an output

' -----[ Program Code ]----------------------------------------------------

Main:

DO ' Main loop

  GOSUB Get_PSX_Packet_Fast ' type and packet

  DEBUG HOME, "Type = "

  IF (psxStatus = $5A) THEN

    DEBUG "psxThumbL = ", DEC psxThumbL, " ",

    "psxThumbR = ", DEC psxThumbR, " "

    DEBUG IHEX2 psxId, "paxStatus (", DEC1 psxStatus, ")",

    CLREOL, CR, CR,

    "ThumbL = ", BIN8 psxThumbL, " ThumbR = ", BIN8 psxThumbR, " "

    IF (psxId <> $41) THEN

      DEBUG " JoyLX = ", DEC3 psxJoyLX, " JoyLY = ", DEC3 psxJoyLY, " ",

      "JoyRX = ", DEC3 psxJoyRX, " JoyRY = ", DEC3 psxJoyRY

    ELSE

      DEBUG CLREOL

    ENDIF

  ELSE

    DEBUG  "Unknown. No response.", "  psxID =  ", DEC psxID, CR, CLRDN

    PAUSE 1000

  ENDIF

  PAUSE 100

LOOP

END ' Main loop

' --- Enter the button values you found below (for your own record) ---

'

' Diagram 1

'

'           L (left)                R (right)

'         _____                   _____

'        |     |                 |     |

'        | [ ] | A               | [ ] | B

'        | [ ] | C               | [ ] | D

'        |_____|                 |_____|

'

'         _____                   _____

'        |     |                 |     |

'        |     |                 |     |

'       /   H   \_______________/   L   \

'      |    O         M     N       O    |

'      | EO   OG     [ ]   [ ]   IO   OK |

'      |    O    ___    [ ]   ___   O    |

'       \___F___/   \____O___/   \__J___/

'               \___/        \___/

'                 P            Q

'

'

'

'

' Button Numbers Chart:

'

' For an indepth explination of this section, see the notes

' section of PS2ControllerConnectToMATLAB.bs2.

'

' **Replace the BIN8 and DEC numbers with the values you found**

'

' psxThumbR:

' A: (L 2),            BIN8 = **11111110**, DEC = **254**

' B: (R 2),            BIN8 = **11111101**, DEC = **253**

' C: (L 1),            BIN8 = **11111011**, DEC = **251**

' D: (R 1),            BIN8 = **11110111**, DEC = **247**

'

' psxThumbL

' E: (L arrow),        BIN8 = **01111111**, DEC = **127**

' F: (down arrow),     BIN8 = **10111111**, DEC = **191**

' G: (R arrow),        BIN8 = **11011111**, DEC = **223**

' H: (up arrow),       BIN8 = **11101111**, DEC = **239**

'

' psxThumbR

' I: (Square),         BIN8 = **01111111**, DEC = **127**

' J: (X),              BIN8 = **10111111**, DEC = **191**

' K: (Circle),         BIN8 = **11011111**, DEC = **223**

' L: (Triangle),       BIN8 = **11101111**, DEC = **239**

'

' psxThumbL

' M: (Select),         BIN8 = **11111110**, DEC = **254**

' N: (Start),          BIN8 = **11110111**, DEC = **247**

' O: (Analog),         Press to activate joystic controles

'

'

' Below is the numbering schemae for the two joysticks (thy both use the same numbering scheme).

' All numbers shown below are in decimal format.

'

'                                    (127, 000)

'                                         ^

'                                         |

' Q: (psxJoyR),                           |

' P: (psxJoyL),     (000, 127)  <--  (127, 127)  -->  (255, 127)

'                                         |

'                                         |

'                                    (127, 255)

' -----[ Subroutines ]-----------------------------------------------------

' Transmit psxOut to, and receive psxIn from the

' PSX controller

PSX_TxRx:

  FOR idx = 0 TO 7

    PsxCmd = psxOut.LOWBIT(idx)                      ' setup command bit

    PsxClk = ClockMode                               ' clock the bit

    psxIn.LOWBIT(idx) = PsxDat                       ' get data bit

    PsxClk = ~ClockMode                              ' release clock

  NEXT

RETURN

' This routine combines manual and built-in shifting

' routines to get the best speed and all valid data.

'

' Execution time on BS2 is ~40 ms.

Get_PSX_Packet_Fast:

  LOW PsxAtt                                       ' select controller

  SHIFTOUT PsxCmd, PsxClk, LSBFIRST, [$01]         ' send "start"

  psxOut = $42 : GOSUB PSX_TxRx                    ' send "get data"

  psxId = psxIn                                    ' save controller type

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxStatus]     ' should be $5A ("ready")

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxThumbL]

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxThumbR]

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxJoyRX]

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxJoyRY]

  SHIFTIN PsxDat, PsxClk, LSBPOST, [psxJoyLX]

  GOSUB PSX_TxRx : psxJoyLY = psxIn

  HIGH PsxAtt                                      ' deselect controller

RETURN


